
Occultations de planètes par la Lune
Ephémérides astronomiques

Serveur de l’IMCCE

Mouvement de la Lune

La position du plan de l’orbite de la Lune inclinée de 5/15 sur l’écliptique, et les variations rapides des
éléments de celle-ci (nœuds, périgée…) font qu’au cours des révolutions successives, la trajectoire apparente de
la Lune sur le ciel se modifie sans cesse.

En quelques années, la Lune a balayée une surface non négligeable du ciel, cachant successivement pendant
quelques minutes à plus d'une heure les objets situés sur sa trajectoire.

Constamment la Lune occulte des étoiles et plus rarement des planètes.
Le phénomène de disparition et de réapparition pour les objets brillants est facile à observer surtout s’il se

produit du côté de l’ombre.
La Lune étant un objet proche, le phénomène observable par effet de parallaxe est variable en position et

temps avec la position de l’observateur.
Grâce aux éphémérides précises topocentriques de l’IMCCE, nous allons pouvoir tracer le mouvement

apparent de l’astre par rapport à la Lune, et prédire avec une bonne précision les positions et les instants de
disparition (Immersion) et de réapparition (Emersion).

Pour les étoiles le phénomène est extrêmement bref. Pour les planètes il dure plusieurs secondes. Il ne sera
pas tenu compte du diamètre apparent de la planète qui allonge le temps d'immersion et d'émersion, mais cela
peut être envisagé.

Les occultations de planètes par la Lune peuvent être étendues aux occultations d'étoiles. Le problème est plus
simple car durant l'occultation, l'étoile n'a pas de mouvement sur le ciel. Mais il faut allez chercher la position
de l'étoile hors IMCCE et être bien compatible en coordonnées.

Ephémérides

L'événement étudié sera l'occultation du 2 mars 2007 à 3h TU.

Nous nous plaçons dans le repère topocentrique en coordonnées équatoriales ou locales (ou horaires).
Les éphémérides donnent à tout instant choisi :

• la direction de la planète α P et δ P (alpha et delta) ou  a P et h P (azimut et hauteur) 
• la direction de la Lune α L et δ L ou  a L et h L 

Nous connaissons :
• le diamètre de la Lune RL

Scénario

Pour suivre l’occultation, nous allons calculer avec les éphémérides, pas à pas :
• θ l’angle entre la direction du centre de la Lune et la direction de la planète.
• ρ L le rayon angulaire de la Lune

Et tracerons les graphes des variations de ces grandeurs en fonction du temps

A partir des données, nous chercherons
• les instants où la planète touche le bord de la Lune
• par le calcul le moment précis de contact
• l'angle de position de la planète relativement à la direction polaire ou zénithale lors des contacts.

Plan de travail

1) Téléchargement depuis le site de l’IMCCE des éphémérides de la Lune et de la planète.
Coordonnées topocentriques du lieu d’observation
Coordonnées équatoriales ou horaires ou locales
Coordonnées Apparentes (équateur vrai ; équinoxe de la date)
De 2 heures avant à 2 heures après par rapport au milieu de l’occultation, pas de 1 minute ou 30 secondes

 2) Transformation des données :
dates, heures
coordonnées topocentiques
distance pour la Lune

3) Calculs : 
Différences de coordonnées et distance angulaire
Rayon angulaire de la Lune
Recherche des contacts et ajustement pour les calcul des instants
Calcul des angles au pôle ou au zénith

4) Tracé des graphiques

Fichier tableur et feuilles de calcul 
 Le stockage et la manipulation des données se fait au moyen d’un tableur, en l’occurrence Excel. Tout est

transposable dans d’autres programmes.
 Pour l’uniformisation des différents fichiers d’éphémérides, chacun comprend

• une feuille d’éphémérides pour chaque objet ([Lune] et [Planète])
• une feuille de calculs et de graphiques ([Occultation]).

Données IMCCE (http://www.imcce.fr)

 Les pages de résultats des éphémérides de l’IMCCE
comportent une partie de mise en page encadrant les
résultats. Les données utiles commencent à la première ligne
de « # » et finissent à la dernière ligne d’éphémérides.

           Entête

         ligne 16 ou 17Partie à sélectionner
à copier, coller

dans les feuilles Excel

 
Toute cette partie est à sélectionner, à copier (CTRL C)

et à coller (CTRL V)  à partir de la cellule A1 de la feuille
d’éphémérides du corps.

La partie éphémérides de l’IMCCE comporte un entête de
renseignements, nom du corps, coordonnées, dates, etc. Cette
partie est à sélectionner aussi, pour ne pas se mélanger dans
les éphémérides téléchargées.

En conséquence, la première ligne d’éphémérides
commence à la ligne 16 ou 17 suivant le type de coordonnées
choisies.

 Pour simplifier la manipulation des les données, il n’y a pas de liens entre les feuilles d’éphémérides de
chaque corps, et les données de la feuille de calculs. Libre à l’utilisateur de le faire.

En conséquence, les éphémérides de la feuille « Occultation » sont à prendre dans les feuilles des corps et à
reporter dans la feuille de « Occultation » avec « copié-collé spécial valeurs ».

Calcul d’éphémérides à l’IMCCE

Pour la Lune et la planète :

Î Un corps du Système solaire  Lune, Soleil

Ï La théorie planétaire garder celle par défaut

Ð Centre du repère topocentrique

Ñ Plan de référence équateur

Ò Type de coordonnées sphériques

Ó Type d’éphémérides Astrométrique J2000

Ô L’échelle de temps  UTC

Õ L’époque du calcul
première date 2007 03 02 02 00 00
pas échantillonnage 30s
nombre de points 500

Ö Lancer le calcul

Entrer à la demande, le département, la ville choisie ou les
coordonnées précise d'un lieu.

Sauvegarde des résultats

Ouvrir Excel et créer 3 feuilles que l’on renommera :
Occultation
Lune
Planète

Pour la  Lune et la planète, faire un copié-collé des données
obtenues sur le site de l’IMCCE dans la colonne A de la feuille
correspondante Excel. 

Pour la partie à  sélectionner voir page précédente.

Note : si dans Excel, le caractère décimal des feuilles de calcul
n'est pas le point décimal vous pouvez soit

- dans un Excel mettre le point comme caractère décimal,
- dans les éphémérides de la colonne A, changer les points

en virgules. 

Fichiers Excel de travail :
Fichier résultat complet  occult_20070302.xls
Fichier structure seule occult_20070302-0.xls

à prendre sur le site 
http://www-obs.univ-lyon1.fr/fc/ateliers_2006-07/ateliers.htm

Page web à remplir pour le calcul

Conversion des données en valeurs numériques
Les formules de calcul s'appliquent aux cellules de la ligne 16. Il suffit ensuite d'appliquer par [sélectionner

glisser] pour l'appliquer aux autres lignes de données, ce qui incrémente le rang des cellules dans les formules.

> Dates et heures ([colonne B et C])
Les dates étant communes aux deux objets, il suffit de les extraire une fois  pour la Lune ([colonnes B et C])

et les reporter dans les colonnes A et B de la "feuille Eclipse" par un collé spécial valeurs.
Dates : formule [cellule B16]

=DATE(CTXT(STXT(A17;9;4));CTXT(STXT(A17;4;3));CTXT(STXT(A17;1;3)))
Heures : formule [cellule C16]

=(CNUM(STXT(A17;14;2))+CNUM(STXT(A17;17;2))/60+CNUM(STXT(A17;20;5))/3600)/24

Ne pas oublier de mettre le format Date dans la [colonne B] et format Heure dans la [colonne C]. Le
formatage  (hh:mm) suffit.

> coordonnées ([feuille Lune et Saturne colonnes D et E])

• Ascensions droites
= CNUM(STXT(A17;27;3))+CNUM(STXT(A17;31;2))/60+CNUM(STXT(A17;34;7))/3600
• Déclinaisons
= (CNUM(STXT(A17;44;2))+CNUM(STXT(A17;47;2))/60+CNUM(STXT(A17;50;7))/3600)*

(SI(STXT(A17;43;1)="-";-1;1))
ou

• Azimuts
=CNUM(STXT(A17;27;3))+CNUM(STXT(A17;31;2))/60+CNUM(STXT(A17;34;7))/3600
• Hauteurs
=(CNUM(STXT(A17;42;2))+CNUM(STXT(A17;45;2))/60+CNUM(STXT(A17;48;7))/3600)*

(SI(STXT(A17;41;1)="-";-1;1))

Données générales

Mettre dans les cellules de la [feuille Lune] :
• [cellule B6] : le rayon de la Lune 1737.4 km
• [Cellule B8] : l'unité astronomique en kilomètre 149597870 km

Dans les formules, pour leur utilisation, il faudra l'écrire avec $ ($C$L) pour éviter les décalages.

Disposition feuille Lune et Saturne – Données tirées des éphémérides

Colonne Feuille Saturne Feuille Lune

A Données IMCCE Données IMCCE

B Dates Dates

C heures heures

D Coordonnées α ou a Coordonnées α ou a

E Coordonnées δ ou h Coordonnées δ ou h

F Distance (unités astronomiques)

[Feuille Lune]

Dans la [feuille Lune colonne G] nous ferons un premier calcul sur tous les rangs, celui du rayon angulaire
de la Lune ρ Lune.

Formule : =DEGRES(ATAN($B$6/($B$8*F17)))

Disposition feuille Eclipse – Données tirées des éphémérides

Colonne Contenu Provenance par 
Copié-collé spécial valeurs

A Dates  [Feuille Lune col. B]

B heures  [Feuille Lune col. C]

D Coordonnées α ou a Saturne  [Feuille Saturne col. D]

E Coordonnées δ ou h Saturne  [Feuille Saturne col. E]

F Coordonnées α ou a Lune  [Feuille Lune col. D]

G Coordonnées δ ou h Lune  [Feuille Lune col. E]

Disposition feuille Eclipse – Données calculées à partir des éphémérides

Col. Contenu Calcul

C heures continues calculées pour le problème de changement de jour

H ∆α cosδ ou ∆ a cos h en minutes d'arc

I ∆δ ou ∆ h en minutes d'arc

J θ distance angulaire, axe du cône Lune en minutes d'arc

K diamètre angulaire de la Lune ρ Lune. en minutes d'arc [Feuille Lune col. G]

M angle de position sur le cercle lunaire
Tracé du cercle lunaire
en minutes d'arcN abscisses des points du cercle lunaire

O ordonnées des points du cercle lunaire

         Colonnes
    M      N          O

Rangs

10
11
12
13
14
15
16
17
18
19
20
...

Dans l'encadré [cellules E3 à G5],seront les résultats :  les calculs des instants des contacts et les angles aux
pôles (ou au zénith). 

Heures continues

Une occultation suivant la distance au centre lunaire peut durer plus d'une heure à quelques instants.
Habituellement, ceci se passe dans une même journée. Mais il arrive que l'occultation commence un jour et
finisse le lendemain.

Dans ce cas, il y a un saut de 24 à 0 heures dans les valeurs de l'heure, ce qui est très gênant pour tracer un
graphe avec le temps comme abscisses. 

Sous Excel les heures sont codées en jours décimaux (0h  0.0, 1h  0.041666, 2h  0.08333, 3h  0.25
etc, mais apparaissent sous la forme conventionnelle hh:mm:ss par le formatage.

Pour éviter le problème au passage à 0h, on ajoute 1 (24h).
Dans la [feuille Occultation] on crée une [colonne C] heure continue :
 S'il y a passage à 24 heures, l'heure continue de croître : 24h, 25h, 26h… mais s’affichera 0h, 1h, 2h… avec

le format heure d’Excel.
Ceci se traduit par la formule  de la [cellule C17] : =(A17-$A$17+B17), à reporter dans toutes les cellules de

la [colonne C]. Cette formule permet de travailler en continu sur plusieurs jours.

Cercle lunaire

Pour mieux visualiser, l’occultation, il y a lieu de tracer sur le graphique du
déplacement relatif de Saturne par rapport à la Lune, le cercle lunaire.

Son diamètre angulaire varie, mais entre le début et la fin de l’occultation, on peut
le considérer comme fixe et prendre sa valeur moyenne.

Repérer les valeurs du rayon angulaire de la Lune [col. K] quand Saturne est occulté
(θ < ρ Lune) et calculer la moyenne dans la [cellule N12] =MOYENNE(K198:K201)).

Sur deux colonne, on va calculer pour des angles β allant de 0 à 360/, les valeurs de 
x = ρ Lune cos β    et    y = ρ Lune sin β 

que l’on tracera sur le graphique.

Construction des points du cercle
[col. M] angles en degré 0, 1, 2, … . 360
[col. N]  x = ρ Lune cos β =$N$12*COS(RADIANS(M17))
[col. O]  x =  ρ Lun sin β =$N$12*SIN(RADIANS(M17))

Que peut-on calculer et tracer

De la Terre on va donc voir la Lune s'approcher de Saturne, la cacher ou l'occulter (Immersion). Quelques
temps après, Saturne va réapparaître (Emersion).

Le bord de la Lune n'étant pas un cercle parfait, le relief lunaire peut provoquer suivant la trajectoire, des
immersions et émersions successives, surtout quand l'occultation est rasante. 

Graphiquement, il va falloir suivre l'évolution dans les temps de la distance angulaire de Saturne au centre
de la Lune. Quand cette distance égale le rayon lunaire, nous avons les instants d'immersion et d'émersion.

Les éphémérides n'étant pas continues, il va falloir rechercher par ajustement ou interpolation, les instants
précis des contacts avec le bord théorique de la Lune.

Pour mieux prévoir les observations, nous calculerons les positions d'immersion et d'émersion soit par rapport
à la direction du Nord (coordonnées équatoriales) soit par rapport au Zénith (coordonnées locales).

Calcul de la distance Centre Lune - Saturne

Calcul des différences des deux coordonnées dans les [colonnes H et I]:
[colonnes H ] (alpha ) ∆α cos δ =(D17-F17)*15*COS(RADIANS(E17))*60

(azimut) ∆a cosh =(D17-F17)*COS(RADIANS(E17))*60
[colonnes I] ∆δ ou ∆h =(E17-G17)*60

L'angle θ se calcule de deux façons :
a) Façon approchée :

 ou θ δ= +( cos )∆α ∆δ2 2 θ = +( cos )∆ ∆a h h2 2

car les angle ∆α et ∆δ sont petits.

Le calcul de θ se fait dans la [colonne J] que l'on exprimera en minutes
d'arc.

Formule de calcul : = RACINE(H17^2+I17^2)

b) Formule rigoureuse : 
cos sin sin cos cos cos( )θ δ δ δ δ α α= ⋅ + ⋅ ⋅ −1 2 1 2 1 2

Tracés

Graphique 1 : évolution de ∆α (ou ∆a , ∆δ (ou ∆h), θ, et ρ Lune en fonction du temps
∆α cos δ ou ∆a cos h [colonne C] versus [colonne H] de 17 à 516
∆δ ou ∆h  [colonne C] versus [colonne  I] de 17 à 516

θ [colonne C] versus [colonne  J] de 17 à 516
   ρLune  [colonne C] versus [colonne K] de 17 à 516

Graphique 2a : déplacement relatif de Saturne par rapport au cercle lunaire : vue d’ensemble
Déplacement : ∆α cosδ versus ∆δ [colonne H] versus [colonne I] (17 à 516)
Cercle lunaire : [colonne N] versus [colonne O] (17 à 377)

Graphique 2b : déplacement relatif de Saturne par rapport au cercle lunaire : vue détaillée
Idem que graphe 2a, avec des limites d’axes sur la partie occultation.

Remarque importante : en coordonnées équatoriales, pour les graphiques 2a et 2b, il faut inverser le sens des
abscisses ∆α, car les ascensions droites croissent de droites à gauche.

Calcul des instants des contacts

On recherchera donc le moment de contact tc par la recherche de
l'intersection de deux droites. L'instant cherché tc sera l'abscisse de
l'intersection.

• Equations des deux droites

  et  y y
t t

y y
t t

a

a

b a

b a

−
−

=
−
−

z z
t t

z z
t t

a

a

b a

b a

−
−

=
−
−

• Abscisse de l’intersection

t t t
z y

y y z z
tb a

a a

b a b a
a= −

−
− − +

+( )

• Ordonnée à l'intersection

y z y
t t
t t

y y z
t t
t t

z zc c a
c a

b a
b a a

c a

b a
b a= = +

−
−

⋅ − = +
−
−

⋅ −( ) ( )

Ici, les deux variables x et y sont θ et ρLune.
Nota : on peut créer une colonne différence θ – ρ Lune et rechercher l'instant de la valeur zéro de cette variable.

L'équation devient : 

, avec  y = 0
y y
t t

y y
t t

a

c a

b a

b a

−
−

=
−
−

t t y t t
y yc a a

b a

b a
= −

−
−

Formule plus facile à transcrire sous excel.

Cellules résultats

Ayant repéré les rangs  où ont lieu les contacts :
Immersion : entre les rangs 197 et 198
Emersion : entre les rangs 201 et 202

on va écrire les formules du calcul des temps dans les
cellules F4 et G4, application directe des formules précédentes.

Les formules des cellules résultats seront :
Immersion [cellule F4] =(C198-C197)*(K197-J197)/(J198-J197-K198+K197)+C197
Emersion [cellule G4] =(C202-C201)*(K201-J201)/(J202-J201-K202+K201)+C201

Angles au pôle ou angle au zénith

Les angles au pôle ou au zénith permettent de mieux situer les positions où auront
lieu l'immersion et l'émersion.

Soient ta et tb, les instants dans les éphémérides qui encadrent le temps de contact
tc, (xa,ya) et (xb,yb) les positions relatives (∆α ou ∆a et ∆δ ou ∆h) de la planète par
rapport à la Lune.

On calcule xc et yc, la position au contact.

 et x x x x
t t

t tc a
b a

b a
c a= +

−
−

−( ) y y y y
t t

t tc a
b a

b a
c a= +

−
−

−( )

L'angle β vaut :  β = −tan 1 y
x

c

c

On en déduit l'angle au zénith : Az = −β 90
ou l'angle au pôle :  Ap = − −360 90( )β

pour tenir compte du sens inverse des ascensions droites.

Formules pour alpha et l'immersion :
xim : =(H198-H197)/(C198-C197)*(P197-C197)+H197
yim : =(I198-I197)/(C198-C197)*(P197-C197)+I197
β : =DEGRES(ATAN2(R197;S197))
P : =360-MOD(DEGRES(ATAN2(R197;S197))+360-90;360)

Pour être sur d'avoir des angles positifs, il faut dans les formules, ajouter 360 et en faire le modulo.
Connaissant les points de contacts, on peut dans les graphiques 2a et 2b, tracer les positions des points de

contacts et matérialiser les angles par des segments de droites (voir fichier excel).

Disposition et programmation

Dans excel faire un encadré spécial de cellules pour mettre les calculs à la hauteur des contacts :

On y inscrira les formules :

Immersion

Calcul Cellule Formule

Temps [P197] =(B198-B197)*(K197-J197)/(J198-J197-K198+K197)+B197

X [R197] =(H198-H197)/(C198-C197)*(P197-C197)+H197

Y [S197] =(I198-I197)/(C198-C197)*(P197-C197)+I197

β [T197] =DEGRES(ATAN2(R197;S197))

angle P [U197] =360-MOD(DEGRES(ATAN2(R197;S197))+360-90;360)
ou =360-MOD(T197)+360-90;360)

Idem pour l'émersion avec les rangs 200 et 201.

Les formules des [cellules P197] et [P201] sont les mêmes que les cellules du tableau de résultats [F5] et
[G5].

Occultation d’étoiles par la Lune

La démarche suivie peut s’appliquer aux occultations d’étoiles.
Le problème est plus simple, l’étoile n’ayant pas de déplacement.

1) Trouver dans un catalogue (Internet ou autre) les coordonnées précises de l’étoile (2000).
2) Les transformer en coordonnées écliptiques (ou travailler en coordonnées équatoriales).
3) Télécharger les éphémérides de la Lune (en système astrométrique 2000).
4) Remplacer les coordonnées de la planète par les coordonnées de l’étoile dans toutes les lignes.

ou
faire disparaître les coordonnées de la planète, mais adapter les formules  de calcul où les coordonnées de

la planète apparaissaient.

Occultation d’étoiles par la Lune : à quoi ça sert ?

Le suivi de la mesure du flux d'une étoile lors d'une occultation revient à observer les franges de diffraction
d'une source quasi ponctuelle par le bord d'un écran. Au lieu d'avoir une image spatiale de la répartition des
franges, il est observé une coupe temporelle du phénomène.

L'espacement des franges est propotionnel à la longueur d'onde observée. C'est pourquoi, les premières
observations furent faites en ondes radiométriques, puis en infrarouge puis en visible à cause des conditions
techniques.

En infrarouge et visible, le phénomène se passe sur quelques centièmes de secondes. Il faut donc un
échantillonnage du signal très rapide, de l'ordre de quelques millième de secondes.

L'intensité des franges et leur espacement sont directement liés à l'étendue de la source.
On peut ainsi mesurer des diamètres angulaires d’étoiles de l’ordre de quelques milliarcsecondes.

Dans les illustrations ci-dessous on voit :
1) deux exemples de signaux d'occultation et l'influence de la taille angulaire de l'objet.
2) de l'analyse de l’espacement et de l’intensité des franges de diffraction, la taille de l'objet avec présence

d’enveloppe étendue dissymétrique.
Illustrations tirée de The Messenger (revue de l’ESO) n/ 126 Dec., 2006 p24-26

